

# Programa para la determinación de hipocentros usando una micro-computadora de bolsillo

Ronald F. Woodman\*

## Resumen

Se ha desarrollado un programa en BASIC para el cálculo de hipocentros en base a los tiempos de arribo de las ondas P a tres estaciones sísmicas, y el tiempo de arribo de una onda S a una de las estaciones, si esta estuviera disponible. El programa, entre código y constantes, usa menos de 1500 caracteres y cabe en una microcomputadora de bolsillo, las que, hoy en día, se pueden conseguir en el mercado por menos de US \$ 150. El programa fue motivado por las necesidades del Instituto Geofísico del Perú, pero tiene una aplicación más universal cuando se considera su portabilidad y costo. Los algoritmos utilizados son descritos en detalle, definiendo en esta forma sus alcances y limitaciones.

## Introducción

Por años la determinación de epicentros regionales y locales en muchos centros del mundo se ha venido haciendo a mano, con el uso de tablas de propagación y métodos gráficos. El servicio de emergencia para la ubicación de epicentros, del IGP, no es excepción. Si bien existen programas con diferente sofisticación (ej. HYP071, FASTHYPO) para la determinación de epicentros por medio de computadoras, en forma automática y más precisa, estas máquinas no siempre están disponibles o accesibles en forma rápida en los centros que tienen la necesidad de efectuar estos servicios. En el caso del Perú, es el Instituto Geofísico del Perú quien tiene la responsabilidad de dar el servicio de emergencia en la determinación de epicentros, y si bien cuenta con dos computadoras capaces de correr programas so-

fisticados y rápidos para esta función, éstas están actualmente físicamente ubicadas en lugares alejados del centro de comunicaciones del Instituto donde se recibe la información sísmica. Por lo tanto, para el servicio de emergencia se hace una determinación de epicentro preliminar a mano. La falta de exactitud de este procedimiento no presenta mayor problema, ya que la idea es tener una ubicación aproximada del epicentro y su magnitud para la toma de decisiones por parte de Defensa Civil respecto a medidas de auxilio. Pero, este servicio puede mejorar significativamente, en precisión, rapidez y confiabilidad si se hace uso de pequeñas micro-computadoras de bolsillo que pueden conseguirse en el mercado por aproximadamente US \$130 dólares, con posibilidad de programación en BASIC y con capacidad de memoria para unos 1,500 caracteres.

El propósito de este trabajo es presentar un programa escrito en BASIC para una de esas micro-computadoras, motivadas por las necesidades del Instituto Geofísico del Perú, pero que pueden ser de utilidad a otros centros que den servicios y tengan problemas similares. El uso de estas computadoras no sólo es capaz de mejorar la precisión de la determinación del hipocentro, sino que, más importante aún, disminuir significativamente el tiempo en encontrarlo. Otro uso potencial del programa es para trabajos de campo en la que la portabilidad y costo de una micro-computadora de bolsillo se hace particularmente útil y deseable. El programa se describe en detalle para que el usuario potencial conozca sus alcances y limitaciones.

## Descripción del programa

El programa ha sido escrito en BASIC para una computadora de bolsillo marca Radio Shack mode-

\* Apartado 3747  
Lima 100-Perú  
Noviembre, 1982. Corregido en Abril, 1983

lo TRS-80. Otros modelos muy similares a ésta en presentación y capacidad son el PC-1211 de la Sharp y EX702P de la Casio. Debido a la universalidad del lenguaje BASIC, el programa puede ser usado fácilmente, también con microcomputadoras de escritorio, aunque se reconoce que éstas serían capaces de aceptar programas más sofisticados.

El programa hace uso del tiempo de arribo de las ondas sísmicas a tres estaciones y requiere solamente los tres tiempos de arribo de la onda P, pero puede hacer buen uso del tiempo de arribo de la onda S a una de las estaciones, mejorando su precisión y disminuyendo el tiempo de cálculo. El programa calcula la latitud y longitud del epicentro, la profundidad del sismo, el tiempo de la ocurrencia y un parámetro de cierre que se puede usar como índice del error de consistencia en el cálculo. Determina también la distancia y el ángulo azimutal con respecto a una de las estaciones (la primera) que haya sido definida.

Los algoritmos utilizados hacen uso de las siguientes suposiciones y simplificaciones: Asume una tierra plana constituida por una capa sobre un semiespacio, cada una caracterizada por velocidades de propagación que satisfacen la relación  $(v_p/v_s)_c = (v_p/v_s)_m$ . Tanto el espesor de la capa superior como el valor de las velocidades en éstas pueden ser asignados para cada zona, cambiando los contenidos de estas variables en tres lugares de memoria.

La latitud y longitud de las tres estaciones, así como los tres tiempos de arribo de la onda P — y el tiempo de arribo de la onda S en la primera estación si este es conocido — son alimentadas por el operador al correrse el programa. El programa solicita estos datos al iniciar la corrida.

El algoritmo utilizado hace uso de un proceso de iteración, empezando por una profundidad fija (20 kilómetros en la lista de la Tabla 1) y un tiempo inicial que corresponde al más tardío posible, consistente con la distancia entre las estaciones y el tiempo de arribo de las ondas. Luego en un proceso de "corte y prueba" se cambia el tiempo en decrementos de un segundo. En cada iteración, se calculan los dos posibles lugares para el epicentro del evento, consistentes con el tiempo de arribo de la onda P en la primera y segunda (1 y 2) estación, ésto es la intersección de las dos circunferencias centradas en las estaciones con radios iguales a la

proyección en la superficie de la distancia, que corresponde a la diferencia entre el tiempo asumido de ocurrencia del sismo y los observados. Se hace lo mismo con las estaciones 1 y 3.

Para calcular este radio, se hace uso del principio de reciprocidad, esto es, que el tiempo y la geometría de propagación es el mismo cuando se excita una onda en el lugar de recepción y se recibe en el de excitación. Si el tiempo y la profundidad son correctos, dos de los puntos correspondientes, uno por cada par de estaciones, coinciden en el mismo lugar y determinan la ubicación del epicentro. Si el tiempo y la profundidad son bastante aproximados esta coincidencia no es exacta, pero la distancia entre estos dos puntos es pequeña y, en todo caso, excluye como posible solución a los otros dos puntos encontrados. Para cada profundidad, y si no se dispone de la Onda S, se varía el tiempo buscando aquel que reduzca la distancia entre las dos soluciones correspondientes a las dos líneas de base (1 y 2, 1 y 3) a un mínimo y se considera éste como el tiempo más aproximado. Luego la profundidad se varía, en forma geométrica multiplicando a la anterior por un factor igual a  $\sqrt{2}$ . Para cada nueva profundidad se repite el procedimiento anterior con tiempos de ocurrencia del sismo cercanos o mayores al tiempo encontrado en la última iteración. Eventualmente la cercanía de las dos soluciones se deteriora, en cuyo caso se regresa a la profundidad y tiempo que produjo la mejor solución; esto es, la menor distancia entre las dos soluciones posibles. En realidad el parámetro que se minimiza es el cuadrado de esta distancia. El punto medio entre las dos soluciones es usado para el cálculo de la longitud y latitud del epicentro.

En caso que se conozca el tiempo de arribo de una onda S, ésta es usada en dos formas: Primero, para estimar el tiempo de la ocurrencia del sismo en forma directa — como lo veremos más adelante — tiempo que es usado para obtener una primera aproximación. Segundo, para calcular la distancia recorrida por la onda S desde la estación 1 que la recibió, para cada tiempo y profundidad, en un proceso de iteración igual al descrito. En este caso, el parámetro minimizado es la suma del cuadrado de la distancia de las dos soluciones de la onda P más el cuadrado de la diferencia entre las distancias evaluadas con la onda P y la S para el tiempo y profundidad considerado. Esto es, ba-

jo un criterio de error mínimo cuadrado, con igualdad de pesos para ambos errores.

En caso que se tenga un mayor número de estaciones, o tiempos de arribo de las ondas S, la información adicional puede ser aprovechada tomando consecutivamente ternas de datos diferentes, que den soluciones independientes, las que pueden ser promediadas de acuerdo a la confiabilidad de las soluciones.

Cabe mencionar que el proceso de minimización descrito para uno u otro caso podría haber sido resuelto por otros métodos clásicos de linealización (Newton, Gauss, etc.) y de convergencia más rápida, pero se decidió seguir el procesamiento descrito de corte y prueba por su simplicidad de implementación. La simplicidad cobra especial importancia cuando se considera que el programa escrito usa todos los lugares de memoria disponibles, no sin antes haber recurrido a una serie de artificios, muchas veces en perjuicio de la claridad del programa.

Los saltos discretos, de un segundo para el tiempo y de un factor de  $\sqrt{2}$  en la profundidad, producen un error de  $\pm 1/2$  segundo en el tiempo, que equivale a 4 km en la determinación del epicentro y de  $\approx 20\%$  ( $\pm 1/2 (\sqrt{2}-1) \times 100$ ) en la determinación de la profundidad. Sin embargo, la mayor fuente de error es la suposición de estratificación horizontal. Es conocido que el espesor de la corteza de los Andes en el Perú varía desde unos 15 km por debajo de la fosa a  $\sim 70$  km por debajo de la cordillera central y oriental. Esta variación de pro-

fundidad no es incluida, aún en programas mucho más sofisticados que usan computadoras de gran capacidad. Por otro lado, la mayor parte de los sismos en el Perú ocurren en el manto, en cuyo caso la solución es poco sensible al modelo, ya que la mayor parte de la propagación, tanto en el modelo o en la realidad, ocurre en este con una velocidad relativamente uniforme y constante. Se recomienda, por lo tanto, usar un modelo con un espesor igual al que tiene la corteza bajo las estaciones sísmicas (digamos 30 km para las estaciones en la costa del Perú) con una velocidad de la corteza correspondiente a la del promedio ponderado de ésta. Para sismos superficiales o redes zonales, este criterio puede cambiar de acuerdo a las propiedades de la corteza en la zona de interés.

#### Algoritmos usados en el cálculo de propagación

El programa considera dos casos (ver fig.: 1) Que el sismo ocurra en la capa superior y, 2) Que el sismo ocurra en el espacio inferior.

En el primer caso se consideran dos modos de propagación: el directo y el refractado. Para el directo la relación distancia, R, tiempo, t, está dada por

$$R = v_c \sqrt{t^2 - H^2/v_c^2} \quad (1)$$

donde  $v_c$  es la velocidad de la onda y H la profundidad del sismo.

Para el refractado,




Figura 1. Micro-computadora de bolsillo modelo TRS-80.




Figura 2. Geometría del método de solución  $\bar{P}$  = Hipocentro,  $P_{12}$  y  $P_{13}$  soluciones aproximadas,  $P_{12}$  y  $P_{13}$  soluciones ambiguas excluidas,  $R$  = distancia al epicentro en base al tiempo de arribo de la onda Pa S1,  $R_s$  = IDEM, pero para la onda S.

$$R = r_c + (t - t_c) v_m \quad \text{para } t > t_c \quad (2)$$

donde  $r_c$  es la suma de la proyección horizontal de los espacios recorridos en la capa superior (ver fig. 3.).

$$r_c = \frac{h' v_c}{\sqrt{v_m^2 - v_c^2}} \quad (3)$$

$t_c$  el tiempo de propagación en la capa superior dado por:

$$t_c = \frac{r_c v_m}{v_c^2} \quad (4)$$

y donde,  $h'$  es la profundidad de la imagen dada por  $2Q - H$ ,  $v_m$  la velocidad en la capa inferior, y  $Q$ , el espesor de la capa superior.

De los dos modos, se toma el más rápido.

Para el segundo caso, existe un sólo modo, para el cual no se puede encontrar una solución explícita de  $R$  en función de  $t$ . Se calcula en su lugar ambas variables paramétricamente en función de  $t_m$ , el tiempo transcurrido en la capa inferior y tenemos:

$$R = r_m + r_c \quad (5)$$

$$t = t_m + t_c \quad (6)$$

donde  $r_m = v_m \sqrt{t_m^2 - (H-Q)^2 / v_m^2}$

$$r_c^2 = Q^2 - \frac{r_m^2 v_c^2}{v_m^4 t_m^2 - v_c^2 r_m^2}$$

y  $t_c = \frac{\sqrt{r_c^2 + Q^2}}{v_c}$




Figura 3. Modelo de estratificación y propagación.

En estas ecuaciones  $r_m$  y  $r_c$  son las distancias horizontales recorridas en las capas inferior y superior, las que están relacionadas por la Ley de Snell de refracción.

Para calcular el valor de R para un tiempo dado t, primero se estima un valor aproximado de  $t_m$  y se calcula el valor exacto de t y R para este tiempo y para otro en su vecindad; el valor deseado para R (t), se obtiene por interpolación lineal usando estos dos puntos cercanos a t. El valor aproximado de  $t_m$  se obtiene suponiendo que el ángulo de propagación en la capa superior es cercano al crítico, esto es que:

$$t_m = t - Q/\nu_c \cos \theta_c = t - \frac{Q}{\nu_c \sqrt{1 - \nu_c^2/\nu_m^2}} \quad (7)$$

Si es que el tiempo es lo suficientemente grande  $t > Q/\nu_c + 2(H-Q)/\nu_m$ . Esto es el epicentro está suficientemente alejado en relación a su profundidad.

Para tiempos menores se asume una relación lineal de  $t_m$  en función de t, tomando dos puntos, uno conocido cuando la propagación es vertical,

$$t_m = \frac{H-Q}{\nu_m} \text{ correspondiente a } t = \frac{t_m + Q}{\nu_c} \text{ y otro}$$

$$\text{para } t_m = \frac{2H-Q}{\nu_m} \text{ correspondiendo a un valor estimado de } t = t_m + \frac{Q}{\nu_c \cos \theta_c}.$$

Aquí se supone

que, para un ángulo de incidencia de  $60^\circ$  en la capa inferior, en la capa superior la dirección de propagación es cercana a la crítica. Esta aproximación no define la precisión del cálculo, pues sólo sirve para encontrar un punto cercano al deseado como base de una expansión lineal en su vecindad.

Como hemos mencionado antes, en el caso de tenerse el tiempo de arribo de la onda S a una de las estaciones, usamos este valor y el arribo de la onda P para estimar el tiempo de ocurrencia del sismo. Es conocido que si la relación de velocidades,  $\gamma$ , entre la onda P y la S, es constante para todos los estratos, los tiempos de propagación correspondientes están en relación inversa a sus velocidades, esto es

$$\frac{T_p}{T_s} = \frac{\nu_s}{\nu_p} = \gamma \quad (8)$$

o, en forma equivalente:

$$\frac{T_p}{T_p - T_s} = \frac{\gamma}{\gamma - 1} \quad (9)$$

Esto nos permite estimar el tiempo de propagación de la onda P,  $T_p$ , en base al valor de  $\gamma$  y la diferencia entre los tiempos  $T_p$  y  $T_s$  leídos del registro. Este estimado, disminuido en un segundo, es usado en la primera iteración de cálculo. El tiempo que se toma como solución, sigue siendo el que minimiza el error anteriormente descrito.

Para calcular las distancias R recorrida por una Onda S, el programa usa los mismos algoritmos que para la P. Pero en lugar de calcular R usando las velocidades que le corresponden a este modo de propagación usa las mismas fórmulas y velocidades de la onda P pero con un tiempo falso  $t_s$  dado por

$$t_s = \frac{v_p}{v_s} t = \gamma t \quad (10)$$

Este artificio de cálculo se basa también en la suposición de que la relación de velocidades de la onda P y de la S es la misma para las dos capas y dadas por la ecuación (8), y permite ahorrar un número sustancial de líneas de código.

#### Instrucciones de uso

- Prender la computadora con el botón ON.
- Apagar y prender con los botones OFF y ON. Esta operación asegura que ésta no haya sido apagada en el modo de "printer" externo.
- Escribir "RUN 100". El número 1 aparecerá en el registro.
- Escribir la longitud de la estación 1, aquella que tenga el tiempo de arribo de la onda S, ó la que tenga el tiempo más tardío. Presionar ENTER.
- Escribir la latitud de la estación 1, luego ENTER
- Escribir el tiempo de arribo de la onda P en segundos con respecto a un minuto de referencia, luego ENTER.
- Escribir lo mismo pero para la onda S, luego ENTER.
- Escribir 0 (cero) en su lugar, si no se dispone de este tiempo.
- El número 2 aparecerá en el registro. Escribir en igual forma la longitud, latitud, tiempo de arribo de la onda P y ENTER.
- El número 3 aparecerá en el registro. Hacer lo mismo para la estación 3. Tomar nota que el programa no toma el tiempo de arribo de la onda S ni para la estación 2 ni la 3.
- Después de unos segundos se escuchará un BEEP y luego cada vez que ocurre una iteración con un nuevo tiempo. Cuando se itera una nueva profundidad se escucha 2 BEEPS.
- Al concluir se escuchan 5 BEEPS la longitud y latitud del epicentro aparecerá en el registro.
- Al presionar ENTER, el tiempo de ocurrencia del sismo en segundos y su profundidad en kilómetros aparecen en el registro.
- Al presionar ENTER nuevamente, aparece el ángulo de azimut y distancia en kms. del epicentro con respecto a la estación 1.

- Si se presiona ENTER una vez más, aparecerá el error de consistencia en kilómetros entre las dos soluciones obtenidas: aquella que usa la información de las estaciones 1 y 2 y la que usa las estaciones 1 y 3.

En caso se escuche 3 BEEPS, significa que los tiempos de arribo no son consistentes con las distancias de las estaciones y la velocidad del modelo. Cuando esto sucede después de haber hecho por lo menos una iteración correcta el programa retorna a la mejor iteración efectuada y da la respuesta. Si lo hace en el inicio, prueba una profundidad menor, por un factor de  $\sqrt{2}$ , si lo encuentra inconsistente repite la operación con la esperanza que una profundidad aún más pequeña de una respuesta consistente. Si la computadora continúa sonando 3 BEEPS discontinuar la corrida a mano.

Si se tiene mayor número de estaciones, a más de un tiempo de arribo para la onda S, se puede repetir el procedimiento tomando otra combinación de estaciones y promediar los resultados. Esto, aparte de mejorar la precisión del epicentro da un mayor grado de confianza en los resultados. Tomar nota también que el programa no es simétrico con respecto a las tres estaciones, pues la No. 1 tiene un tratamiento diferente, y una rotación de las estaciones puede ser usado también con funciones similares.

El programa hace uso de las tres "constantes" que deben ser iniciados, éstas son los valores de J, P, Q correspondientes a la velocidad de la capa superior, la de la inferior y (en  $^{\circ}/\text{sec}$ ) el espesor de la capa en grados (kms. x 0.009).

El programa puede ser modificado también para considerar otra profundidad inicial, H, y relación de velocidades  $\gamma_p / \gamma_s$ . La primera está definida en la línea de instrucción 140: H=0.18 (correspondiente a 20 km x 0.009  $^{\circ}/\text{km}$ ), la segunda en 63:... T= .577 en donde

$$\gamma = \frac{v_s}{v_p} = .577 \quad (17)$$

y también en la línea 160: ... A=2.36 A(27)-1.36A(30) + 1, en donde los coeficientes corresponden a

$$\beta + 1 \text{ y } \beta = \frac{\gamma}{1 - \gamma} \text{ respectivamente. Al cambiar}$$

estos coeficientes hay que cuidar que el número total de dígitos no sea mayor pues el programa ya usa todos los lugares disponibles de memoria.

**TABLA 1**  
**LISTADO DE PROGRAMA**

**Programa Principal**

```

100: INPUT "1", A(31), A(34), A(27), A(30), "2", A(32), A(35), A(28), "3", A(33), A(36), A(29)
120: Y=A(34)-A(35): Z=A(31)-A(32): GOSUB 56: I=D:A(37)=L
130: Y=A(34)-A(36): Z=A(31)-A(33): GOSUB 56: K=D:A(38)=L
140: H=.18:A=A(27)+A(28)-A(37)/P:X=A(27)+A(29)-A(38)/P: IF X<A LET A=X
150: A=A/2
160: Z=10^4:A(33)=Z:A(35)=0:A(36)=Z: IF A(30)>O LET A=2.36A(27)-1.36A(30)+1
170: A(32)=A: V=PP: W=JJ: C=QQ: N=Q/J: O=N/√(1-W/V): GOTO 34
34 : T=A(27)-A: GOSUB 1: B=R: T=A(28)-A: GOSUB 1: F=R: T=A(29)-A: GOSUB 1: G=R
40 : Y=B: Z=F: L=A(37): GOSUB 48: F=D: IF D<-7 GOTO 67
42 : Z=G: L=A(38): GOSUB 48: G=D: IF D<-7 GOTO 67
46 : L=9999: U=I+F: E=K+G: GOSUB 59: E=K-G: GOSUB 59: U=I-F: E=K+G: GOSUB 59: GOTO 62
62 : IF A(35)=1 GOTO 76
63 : IF A(30)>O LET T=.5777*(A(30)-A): F=L: GOSUB 1: L=R-B: L=F+LL
66 : IF L>A(36) GOTO 70
67 : IF D=-9 BEEP 3: GOTO 74
68 : BEEP 1: A=A-1: A(36)=L: GOTO 34
70 : IF A(36)>A(33) GOTO 74
72 : H=H*√2:A(33)=A(36): A(36)=10^4:A(32)=A+1: A=A+1: BEEP 2: GOTO 34
74 : H=H/√2:A=A(32): A(35)=1: GOTO 34
76 : Y=A(34)-Y: X=A(31)-X: BEEP 5: PRINT X, Y: PRINT A, H/.009: PRINT 57.3U, 111B: PRINT √L*111

```

Subrutina para evaluar R en función de T

```

1 : IF<H =QGOTO 24
2 : X=H-Q: E=XX: M=X/P: IF T<N+M LET R=0: RETURN
3 : IF T<(O+2M) GOTO 10
4 : X=T-O: GOSUB 7: S=Z: G=D
5 : X=X+1: GOSUB 7
6 : R=G+(T-S)*(D-G)/(Z-S): RETURN
10 : X=M+(T-N-M)/M/(M+O-N): GOSUB 7
20 : S=Z: G=D
22 : X=X+M/10: GOSUB 7: GOTO 6
24 : E=HH: M=(2Q-H)J/√(V-W): S=MP/W: IF JT<H LET R=0: RETURN
26 : L=√(WTT-E): R=0
28 : IF T>S LET R=(T-S)P+M
30 : IF R<L LET R=L
32 : RETURN

```

Subrutina 7: R y T de  $t_m$

```

7 : Y=XX: D=YV-E
8 : L=CWD/(VVY-WD)
9 : Z=(L+C)/W: Z=X+√Z: D=√L + √D: RETURN

```

Subrutina para evaluar el ángulo de un triángulo de lados conocidos

```

48 : D=(YY-ZZ+LL)/((Y+1E-7)*2L): IF ABS D>1 LET D=-8: IF ABS (Y-Z)>L LET D=-9
50 : IF D>-7 LET D=ACS D
51 : RETURN

```

Subrutina ATN de cuatro cuadrantes

```

56 : D=ATN (Y/Z): IF Z<O LET D=D+π
58 : L=√(YY+ZZ): RETURN

```

Subrutina para calcular la mejor solución del epicentro

```

59 : M=COS U-COS E: S=SIN U-SIN E: D=(MM+SS)BB
60 : IF D<L LET L=D: IF A(35)=1 LET X=(COS U+COS E)B/2: Y=(SIN U+SIN E)B/2
61 : RETURN

```